多段階選抜 (MoonScript)


|多段階選抜|解答日|シリーズ:yieldの練習/ジェネレータを入れ子に/整数平方根・立方根の実装|
|:--|:--|:--|
|問題|　|http://nabetani.sakura.ne.jp/hena/ord24eliseq/<br>https://qiita.com/Nabetani/items/1c83005a854d2c6cbb69|
|Ruby|2014/8/2(当日)|https://qiita.com/cielavenir/items/9f15e29b73ecf98968a5|
|C#/Python|2014/8/4|https://qiita.com/cielavenir/items/a1156e6a4f71ddbe5dcb|
| |　|ここから上はdrop_prev_square/drop_prev_cubicをまとめる前の答案|
|Go/C#/Ruby/Python|2014/8/5|https://qiita.com/cielavenir/items/2a685d3080862f2c2c47|
|PHP/JavaScript|2014/9/9|https://qiita.com/cielavenir/items/28d613ac3823afbf8407|
|VB|2014/9/10|https://qiita.com/cielavenir/items/cb7266abd30eadd71c04|
|D|2015/12/21|https://qiita.com/cielavenir/items/47c9e50ee60bef2847ec|
|Perl|2017/3/10|https://qiita.com/cielavenir/items/6dfbff749d833c0fd423|
|Lua|2017/3/13|https://qiita.com/cielavenir/items/c60fe7e8da73487ba062|
|C++20(TS)|2017/3/15|https://qiita.com/cielavenir/items/e1129ca185008f49cbab (MSVC)<br>https://qiita.com/cielavenir/items/1cfa90d73d11bb7dc3d4 (clang)|
|F#|2017/3/17|https://qiita.com/cielavenir/items/a698d6a26824ff53de81|
|Boo/Nemerle|2017/5/13|https://qiita.com/cielavenir/items/e2a783f0fe4b0fe0ed48|
|Perl6|2017/5/15|https://qiita.com/cielavenir/items/656ea17fa96c865c4498|
|Kotlin|2017/5/25|https://qiita.com/cielavenir/items/9c46ce8d9d12e51de285|
|Crystal|2018/5/8|https://qiita.com/cielavenir/items/1815bfa6a860fd1f90db|
|MoonScript|2018/6/16|https://qiita.com/cielavenir/items/8b03cce0386f4537b5ad|
|Julia/Rust|2018/12/20|https://qiita.com/cielavenir/items/3ddf72b06d625da0c4a5|
|Nim|2018/12/26|https://qiita.com/cielavenir/items/5728944867e609fd52a7|
|Tcl|2018/12/31|https://qiita.com/cielavenir/items/76cbd9c2022b48c9a2c9|
|Pascal/Cobra|2019/1/16|https://qiita.com/cielavenir/items/81b81baf8dfc1f877903|
|Icon|2019/1/17|https://qiita.com/cielavenir/items/889622dcc721f5a4da24|
|Swift|2020/5/31|https://qiita.com/cielavenir/items/3b0b84a218e35d538f7f|
|Java/Groovy/Scala|2020/5/31|https://qiita.com/cielavenir/items/7f058203a8fd03b65870|
|V|2020/10/17|https://qiita.com/cielavenir/items/df30a6c101a97a713df5|
|Zig/Zen|2020/10/17|https://qiita.com/cielavenir/items/9cced9e4a94dcd70df0f|
|Pike|2020/11/2|https://qiita.com/cielavenir/items/3a8248f41611302b34fd|
|Vala/Smalltalk|2020/11/29|https://qiita.com/cielavenir/items/085dabe593cd916af5e8|
|Objective-C|2020/11/30|https://qiita.com/cielavenir/items/a1736e38789a3dd5cc5a|
|Ruby(Ractor)|2021/1/2|https://qiita.com/cielavenir/items/f493c6d512b63cc571cc|
|Python(_xxsubinterpreters)|2021/6/29|https://qiita.com/cielavenir/items/f1f581a055db918954f1|
|Falcon/Scheme|2021/9/5|https://qiita.com/cielavenir/items/c13d12cf44f0d17f4a94|
|Clojure/Lisp|2021/9/7|https://qiita.com/cielavenir/items/7458ba076f4e5bc0f196|
|(icbrtの実装に関する)補題|2017/5/11|整数除算であってもn/(x*y)はn/x/yに等しい(ことの証明)<br>https://qiita.com/cielavenir/items/21a6711afd6be8c18c55|

処理系 `luarocks install moonscript`

Luaのまるっと移植。[MoonScriptでのAtCoder Beginners Selection踏破](https://qiita.com/cielavenir/items/6553531e230f39cd6a3d)記念です。

```moon:tyama_hena24_enum.moon
#!/usr/bin/env moon
-- http://qiita.com/Nabetani/items/1c83005a854d2c6cbb69
-- http://nabetani.sakura.ne.jp/hena/ord24eliseq/

isqrt=(n)->
	if n<=0
		return 0
	if n<4
		return 1
	x=0
	y=n
	while x!=y and x+1!=y
		x=y
		y=(n//y+y)//2
	x

icbrt=(n)->
	if n<0
		return -icbrt(-n)
	if n==0
		return 0
	if n<8
		return 1
	x=0
	y=n
	while x~=y and x+1~=y
		x=y
		y=(n//y//y+y*2)//3
	x

generate=->
	coroutine.create(->
		i=1
		while true
			coroutine.yield(i)
			i=i+1
	)

drop_prev=(check,prev)->
	coroutine.create(->
		_,a=coroutine.resume(prev)
		_,b=coroutine.resume(prev)
		while true do
			if not check(b)
				coroutine.yield(a)
			a=b
			_,b=coroutine.resume(prev)
	)

drop_next=(check,prev)->
	return coroutine.create(->
		_,a=coroutine.resume(prev)
		_,b=coroutine.resume(prev)
		coroutine.yield(a)
		while true
			if not check(a)
				coroutine.yield(b)
			a=b
			_,b=coroutine.resume(prev)
	)

drop_n=(check,n,prev)->
	return coroutine.create(->
		i=0
		while true
			i=i+1
			_,a=coroutine.resume(prev)
			if not check(i,n)
				coroutine.yield(a)
	)

is_sq=(n)->
	-- x=math.floor(math.sqrt(n))
	x=isqrt(n)
	return x*x==n

is_cb=(n)->
	-- x=math.floor(math.cbrt(n))
	x=icbrt(n)
	return x*x*x==n

is_multiple=(i,n)->i%n==0

is_le=(i,n)->i<=n

f={
	['S']:(e)->drop_next(is_sq,e),
	['s']:(e)->drop_prev(is_sq,e),
	['C']:(e)->drop_next(is_cb,e),
	['c']:(e)->drop_prev(is_cb,e),
	['h']:(e)->drop_n(is_le,100,e),
}

-- ラムダ周りのカッコは必要(forがラムダの中にかかってしまう)
f[tostring(i)]=((e)->drop_n(is_multiple,i,e)) for i=2,9

line=io.read('*line')
while line!=nil
	g=generate()
	g=f[c](g) for c in line\gmatch('.')
	for i=1,10
		io.write(',') if i>1
		_,r=coroutine.resume(g)
		io.write(r)
	io.write("\n")
	io.flush()
	line=io.read('*line')
```